
この発表では定理証明支援系 Lean で Gödel の不完全性定理と証明可能性論理に関する事実を
形式化したということを発表します．厳密に話すと時間がいくらあっても足りないので，余談
を挟みつつラフに解説していきます（スライドも飛ばし飛ばしで使います）．

• 最新版のスライド: https://sno2wman.github.
io/slides-for-MGITaPL/main.pdf

• プレプリント: https://arxiv.org/abs/2609.
13780

• 実装: https://github.com/FormalizedFormalL
ogic/Foundation

注意：プレプリント出版時とリポジトリの構成が大きく異なっています（Foundation へ統合
しています）

https://sno2wman.github.io/slides-for-MGITaPL/main.pdf
https://sno2wman.github.io/slides-for-MGITaPL/main.pdf
https://arxiv.org/abs/2609.13780
https://arxiv.org/abs/2609.13780
https://github.com/FormalizedFormalLogic/Foundation
https://github.com/FormalizedFormalLogic/Foundation

要約

数学の実践において，人はしばしば論理の運行を間違える．例: 仮定の不足，帰納法，etc.

→ 機械に検証させればいいじゃん！
定理証明支援系とは数学をプログラミングコード（形式言語）として記述し，その記述が論理
的に破綻してないことを検証するソフトウェアである．

例: Rocq, Isabelle, Agda, HOL Light, Lean

2020 年に新しく開発された定理証明支援系．

Lean 4 は型理論として Calculus of Inductive (𝖢𝖨𝖢) を採用しており， 原理上は1普通に行われ
ている数学が全部展開出来るだろうとされている．

特徴
• 強力なメタプログラミング能力（マクロや新しいタクティクの作成が容易）．
• 汎用的な数学的ライブラリ Mathlib（代数，解析学，圏論，etc.2）．
• 実用的なプログラミング言語への志向．

近年 AI/LLM が寄越してきた証明の保証として用いられる風潮がある3．

1Lean 3 では Carneiro (2019) が 𝖹𝖥𝖢 +「𝜔-個の到達不能基数が存在する」 の無矛盾性をモデルを作って示している． この結果を素直にこの結果を
Lean 4 に持ってくることは出来ない（らしい）が，最近ようやく https://github.com/leanprover/con-leche などで取り組まれているよう
に思える． もちろん今ある数学が全然 𝖹𝖥𝖢 でやってるわけねーだろという方にとっては一切この話は関係ない．

2学部レベルの数学は全部あるらしい．

3なぜなのかはよくわからない．コンパイルの速度？

https://github.com/leanprover/con-leche

Gödel の不完全性定理の形式化の先行研究として例えば以下がある．

• Shankar (1986) による Boyer–Moore 定理証明器（Nqthm）
‣ G1 の最初の形式化．

• O'Connor (2005) による Rocq での G1, G2 の形式化．
‣ G1 は 𝖯𝖠 ではなくて 𝖹𝟤 という算術体系上で, G2 はいくつかの仮定を公理として．

• Paulson (2014) による G2 の形式化
‣ 唯一の G2 の形式化だが，遺伝的有限集合論で議論．

１つ目の新規性：フルの算術で全部やった．

様相論理 𝐆𝐋 の形式化として
• Maggesi and Perini Brogi (2023) でラベル付きシークエント計算の HOL Light での形式

化．

２つ目の新規性：実際に 𝐆𝐋 の様相を算術に関連付けた形式化は誰もやっていない．

余談: 論理学の形式化はそれほど活発に行われていない！1

1なぜかはよくわからない．単に数学基礎論がそんなに流行ってないという分野の問題かもしれないし，方向性としてこの分野の研究者が伝統的な
記号論理よりも型理論の方に行ってしまうからなのかもしれない．

最近のニュース: Ramana Kumar1 は Collatz 予想の反証を Lean v4.32.1 で形式化した．

-- `n` はCollatzの操作を何度行っても1にならない．
def Diverges (n : Nat) : Prop := 0 < n ∧ ∀ k, iterate step k n ≠ 1

-- そのような `n` が存在する．
theorem exists_nonterminating_orbit : ∃ n, Diverges n :

もちろん2このような上手い話があるわけがなく，これは Lean のソフトウェアとしての実装の
バグに由来するものであった（詳しい解説は Moura (2026)，v4.33.0 で修正）．

甚大な量の Lean による形式証明のソースコードの中にこのような秘孔が混じっている可能性
はある． これに関しては Section 5.5 でも論ずる．

1https://github.com/xrchz/CollatzLean

2この講演が行われた当時は少なくとも

https://github.com/xrchz/CollatzLean

ひとくちに定理証明支援系を使うと言っても，次のことに目を向けるべきだと感じる1．

注意 1 (定理証明支援系による正しさの公準):  定理証明支援系が/の/で．．．
1. 依拠している数学的基礎に問題や矛盾が無いか？（論理学者として）
2. ソフトウェアとしての実装にバグが存在しないか？（ソフトウェア開発者として）
3. 形式化したそのコードは命題を正しく形式化しているか？（エンドユーザとして）

1この基準は @Alwe_Logic (2026) のツイートから拝借した．

注意 2:  formalize という語の意図を以下で使い分ける．

formalize (形式化) 数理論理学の技法として数学的議論を形式操作だと思って展開する．
mechanize (機械化) 定理証明支援系によって数学をプログラムによって実装する．

我々がやったことの端的な説明：数学の形式化の機械化！



3. 1 階述語論理および不完全性定理
3.2 不完全性定理 12/92

次の Gödel の不完全性定理の素朴なバージョンを機械化した．

定理 (Gödel の第 1 不完全性定理(G1)):  𝑇  が Cobham の最弱の算術 𝖱𝟢 を含み，Δ1-定義可
能1 で，Σ1-健全なら，𝑇  から証明も反証も出来ない文が存在する．

定理 (Gödel の第 2 不完全性定理(G2)):  𝑇  が 𝖨Σ1 より強く，Δ1-定義可能で，無矛盾なら，𝑇  
の無矛盾性を表す文は証明できない．

いくつかの条件は改良できる(後述)．

theorem incomplete (T : ArithmeticTheory) [T.Δ₁] [𝗥₀ ⪯ T] [T.SoundOnHierarchy 𝚺 1] : Incomplete T

theorem consistent_unprovable [Consistent T] : T ⊬ T.consistent.val

Source: 1 2

1𝑇 の公理を記述する論理式が Δ1-論理式で記述出来る

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/First.lean
https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/Second.lean

論理式 (擬論理式) は否定標準形で扱う．L は言語，型 ξ を自由変数，束縛変数は de Bruijn イ
ンデックスによって自然数 ℕ で扱うこととする．

𝜑, 𝜓 ⩴ ⊤ | ⊥ | 𝑅(𝑣)⃗ | ¬𝑅(𝑣)⃗ | 𝜑 ∧ 𝜓 | 𝜑 ∨ 𝜓 | ∀𝜑 | ∃𝜑

  inductive Semiformula (L : Language) (ξ : Type*) : ℕ → Type _ where
  |  verum : Semiformula L ξ n
  | falsum : Semiformula L ξ n
  |    rel : {arity : ℕ} → L.Rel arity → (Fin arity → Semiterm L ξ n) → Semiformula L ξ n
  |   nrel : {arity : ℕ} → L.Rel arity → (Fin arity → Semiterm L ξ n) → Semiformula L ξ n
  |    and : Semiformula L ξ n → Semiformula L ξ n → Semiformula L ξ n
  |     or : Semiformula L ξ n → Semiformula L ξ n → Semiformula L ξ n
  |    all : Semiformula L ξ (n + 1) → Semiformula L ξ n
  |    exs : Semiformula L ξ (n + 1) → Semiformula L ξ n

Source: 1

• Formula L ξ を Semiformula L ξ 0 の略記（束縛変数無し）
• Semisentence L n を Semiformula L Empty n の略記（自由変数無し）
• Sentence L を Formula L Empty の略記（自由・束縛変数無し）

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Syntax/Classical/Formula.lean

算術の言語 ℒ︀OR を定めて，Lean のマクロによる糖衣構文を用意する． 例えばこんな感じで記
述出来る．

論理式 Lean

98 + 6748 = 6846 “98 + 6748 = 6846”

∀𝑥. ∀𝑦. 𝑎 ⋅ (𝑥 + 𝑦) = 𝑎 ⋅ 𝑥 + 𝑎 ⋅ 𝑦 “a. ∀ x y, a * (x + y) = a * x + a * y”

∀𝑥. [𝑥 < 𝑛 ↔ ⋁
𝑖<𝑛

𝑥 = 𝑖] “∀ x, x < ↑n ↔ ⋁ i < n, x = ↑i”

ここで a. は束縛変数（ArithmeticSemisentence 1 の変数），↑n はメタの自然数 n の数項である．

古典 1 階述語論理の Tait 流のシークエント計算体系 𝐋𝐊 を用意する．シークエントは多重集合
として定義する．

id
𝑅(𝑣)⃗, ¬𝑅(𝑣)⃗, Δ

verum
⊤, Δ

Γ
wk1

Δ
Γ, 𝜑, 𝜑

ctr
Γ, 𝜑

𝜑, Δ ¬𝜑, Δ
cut

Δ
𝜑, 𝜓, Δ

or
𝜑 ∨ 𝜓, Δ

𝜑, Δ 𝜓, Δ
and

𝜑 ∧ 𝜓, Δ
𝜑+(&0), Δ+

all2
∀𝜑, Δ

𝜑(𝑡), Δ
ex

∃𝜑, Δ

inductive LK.Derivation : LK.Sequent L → Type _
| identity (r : L.Rel k) (v) : LK.Derivation ⦃.rel r v, .nrel r v⦄
| verum : LK.Derivation ⦃⊤⦄
| weakening : LK.Derivation Γ → LK.Derivation (Γ + ⦃φ⦄)
| contraction : LK.Derivation (Γ + ⦃φ, φ⦄) → LK.Derivation (Γ + ⦃φ⦄)
| cut : LK.Derivation (Γ + ⦃φ⦄) → LK.Derivation (Δ + ⦃∼φ⦄) → LK.Derivation (Γ + Δ)
| or : LK.Derivation (Γ + ⦃φ, ψ⦄) → LK.Derivation (Γ + ⦃φ ⋎ ψ⦄)
| and : LK.Derivation (Γ + ⦃φ⦄) → LK.Derivation (Γ + ⦃ψ⦄) → LK.Derivation (Γ + ⦃φ ⋏ ψ⦄)
| all : LK.Derivation (Γ⁺ + ⦃φ.free⦄) → LK.Derivation (Γ + ⦃∀¹ φ⦄)
| exs : LK.Derivation (Γ + ⦃φ/[t]⦄) → LK.Derivation (Γ + ⦃∃¹ φ⦄)

Source: 1

1Γ ⊆ Δ
2ここでは自由変数を &0, &1, … と表記する．また 𝜑+, Γ+ はそれぞれに含まれる自由変数をインクリメントしたもの

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/LK/Basic.lean

カット無しの証明図へ変換する具体的な計算手続きを定める1ことで，𝐋𝐊 ではカット除去定理
を機械化出来る2．

def hauptsatz {Γ : LK.Sequent L} : ⊢ᴸᴷ¹ Γ → {d : ⊢ᴸᴷ¹ Γ // LK.Derivation.IsCutFree d}

Source: 1

言語 𝐿 の理論 𝑇  を 𝐿-文の集合 Set (Sentence L) とする．

• 𝜑 が 𝑇  から導出できることを 𝑇 ⊢ 𝜑 と書く．
• 𝑇  を満たす任意のモデル 𝑀  で 𝜑 も満たされるとき，𝑇 ⊨ 𝜑 と書く．

カット除去定理からカノニカルモデルを作るなどの議論を行って，完全性定理を得る．

theorem small_satisfiable_of_consistent : Consistent T → Satisfiable T

theorem Proof.complete_iff : T ⊨ φ ↔ T ⊢ φ := ⟨fun h ↦ Proof.complete h, Proof.sound⟩

Source: 1

1証明図の帰納法による愚直な証明は機械化において煩雑で面倒なので，Avigad (2004), Avigad (2001) による直観主義述語論理への還元および強
制法的な議論による．

2ただしこれが現実的に Lean で計算可能なのかはわからない．

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/LK/Hauptsatz.lean
https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/LK/Completeness/CounterModel.lean

以降，言語 ℒ︀OR (ℒₒᵣ) の理論を算術と呼ぶ．

定義 3:  Cobham の最弱の算術 𝖱𝟢 は ℒ︀OR の等号公理と，次の公理図式からなる理論である
（𝑛, 𝑚 ∈ ℕ）． 𝑛 は数項を表す．

𝑛 + 𝑚 = 𝑛 + 𝑚, 𝑛 ⋅ 𝑚 = 𝑛 ⋅ 𝑚, ∀𝑥. [𝑥 < 𝑛 ↔ ⋁
𝑖<𝑛

𝑥 = 𝑖]

inductive R0 : ArithmeticTheory
  | equal        : ∀ φ ∈ 𝗘𝗤 ℒₒᵣ, R0 φ
  | Ω₁ (n m : ℕ) : R0 “↑n + ↑m = ↑(n + m)”
  | Ω₂ (n m : ℕ) : R0 “↑n * ↑m = ↑(n * m)”
  | Ω₃ (n : ℕ)   : R0 “∀ x, x < ↑n ↔ ⋁ i < n, x = ↑i”

notation "𝗥₀" => R0

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Arithmetic/R0/Basic.lean

定義 4:  𝖯𝖠− は離散順序半環の基本的な性質を述べる有限個の Π1 文からなる理論である．

abbrev addZero       := “∀ x, x + 0 = x”
abbrev addAssoc      := “∀ x y z, (x + y) + z = x + (y + z)”
abbrev addComm       := “∀ x y, x + y = y + x”
abbrev addEqOfLt     := “∀ x y, x < y → ∃ z <⁺ y, x + z = y”
abbrev zeroLe        := “∀ x, 0 ≤ x”
abbrev zeroLtOne     := “0 < 1”
abbrev oneLeOfZeroLt := “∀ x, 0 < x → 1 ≤ x”
abbrev addLtAdd      := “∀ x y z, x < y → x + z < y + z”
abbrev mulZero       := “∀ x, x * 0 = 0”
abbrev mulOne        := “∀ x, x * 1 = x”
abbrev mulAssoc      := “∀ x y z, (x * y) * z = x * (y * z)”
abbrev mulComm       := “∀ x y, x * y = y * x”
abbrev mulLtMul      := “∀ x y z, x < y ∧ 0 < z → x * z < y * z”
abbrev distr         := “∀ x y z, x * (y + z) = x * y + x * z”
abbrev ltIrrefl      := “∀ x, x ≮ x”
abbrev ltTrans       := “∀ x y z, x < y ∧ y < z → x < z”
abbrev ltTri         := “∀ x y, x < y ∨ x = y ∨ x > y”

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Arithmetic/PeanoMinus/Basic.lean

定義 5:  1 変数論理式 𝜑(𝑥) の帰納法 𝜑(0) → ∀𝑥. [𝜑(𝑥) → 𝜑(𝑥 + 1)] → ∀𝑥. 𝜑(𝑥) の全称閉包を 
Ind(𝜑) と書く．
• 𝖨Σ1 ≔ 𝖯𝖠− + {Ind(𝜑) : 𝜑 は冠頭標準形の Σ1-論理式}．
• 𝖯𝖠 ≔ 𝖯𝖠− + {Ind(𝜑) : 𝜑 は任意の論理式}．

def succInd {ξ} (φ : Semiformula L ξ 1) : Formula L ξ :=
  “!φ 0 → (∀ x, !φ x → !φ (x + 1)) → ∀ x, !φ x”
def InductionScheme (Γ : Semiformula L ℕ 1 → Prop) : Theory L :=
  { ψ | ∃ φ : Semiformula L ℕ 1, Γ φ ∧ ψ = .univCl (succInd φ) }

abbrev InductionOnPrenexHierarchy (Γ : Polarity) (s : ℕ) : ArithmeticTheory :=
  𝗣𝗔⁻ ∪ InductionScheme ℒₒᵣ (ℬ[<, ℒₒᵣ].PrenexHierarchy Γ s)
prefix:max "𝗜𝗡𝗗 " => InductionOnPrenexHierarchy

abbrev ISigma (s : ℕ) : ArithmeticTheory := 𝗜𝗡𝗗 𝚺 s
notation "𝗜𝚺₁" => ISigma 1

abbrev Peano : ArithmeticTheory := 𝗣𝗔⁻ ∪ InductionScheme ℒₒᵣ Set.univ
notation "𝗣𝗔" => Peano

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Arithmetic/Schemata.lean

ここについての扱いは現在改修中．

現れる量化子が限定量化であるような論理式は Δ0-論理式と呼ぶ．

実用的な問題のため，論理式が Σ1 や Π1 であるとは限定量化を途中に挟んでもよく，また
Boolean 結合で閉じているものとする．例えば以下も Σ1-論理式である．

∃𝑥, ∀𝑦 < 𝑡, ∃𝑧.𝜑(𝑥, 𝑦, 𝑧) ∧ ∀𝑥 < 𝑡.𝜓(𝑥)

一方で，帰納法のクラスなどを Σ𝑛 や Π𝑛 で制限するときは冠頭標準形に制限する．

∃∀∃∀…𝜑(𝑥1, 𝑥2, …, 𝑥𝑛, 𝑦)⃗

冠頭標準形定理および採集原理などがあればこれらの区別は厳密にする必要はないが，機械化
ではこのような細かい議論や定義も必要になる．

𝑇 ⊢ 𝜑 であることを（特に Lean で機械化するには）実際に証明図を構成しなければならない
が，それは人力では現実的ではない．

完全性定理を機械化することで以下の還元が使える．

𝖫𝖾𝖺𝗇 ⊢ "𝑇 ⊢ 𝜑" ⟺ 𝖫𝖾𝖺𝗇 ⊢ "∀𝑉 , 𝑉 ⊨ 𝑇 ⟹ 𝑉 ⊨ 𝜑"

意味論的な議論においては，例えば 𝑇  が十分に豊かな算術であるなら 𝑉 ⊨ 𝑇  を満たす 𝑉  が良い
代数的な構造になる． Mathlib などが提供する代数的な構造に対しての様々な補題やメタプロ
グラミング，自動証明タクティクが利用出来る． これを戻して 𝑇 ⊢ 𝜑 を簡単に示せる．

算術 𝑇  の任意のモデル 𝑉  を固定する．

variable {V : Type*} [ORingStructure V] [V↓[ℒₒᵣ] ⊧* T]

• ORingStructure V: 𝑉  が言語 ℒ︀OR の構造であることを主張する typeclass.
• V↓[ℒₒᵣ] ⊧* T: 𝑉  が理論 𝑇  を満たすことを主張する typeclass.

𝑉  上で機械化を行う．関数は選択関数を用いて定義出来る．

lemma sqrt_exists_unique (a : V) : ∃! x, x * x ≤ a ∧ a < (x + 1) * (x + 1)

noncomputable def sqrt (a : V) : V := Classical.choose! (sqrt_exists_unique a)
prefix:75 "√" => sqrt

lemma sqrt_mul_self (a : V) : √(a * a) = a

形式体系を算術の中でさらに形式化する1：算術化・Bootstrapping．

• 論理式 𝜑 や導出木 𝐷 に対して 𝑉  への割り当て ⌜⋅⌝ を割り当てる（Gödel 数）．
• 逆に 𝑉  の要素 𝑥 が項，論理式，導出木の Gödel 数であるというメタの（Lean 上の）述語 

IsFormula(𝑥) : 𝑉 ↦ 2 などを考える．

メタの述語 IsFormula(𝑥) に対応する ℒ︀OR-論理式 𝖨𝗌𝖥𝗈𝗋𝗆𝗎𝗅𝖺(⌜𝜑⌝) などを Δ0, Σ𝑛, Π𝑛 の適当な階層
でひたすら頑張って構成していく．

𝑉 ⊨ IsFormula(⌜𝜑⌝) ⟺ 𝑇 ⊢ 𝖨𝗌𝖥𝗈𝗋𝗆𝗎𝗅𝖺(⌜𝜑⌝)

1Formalizing [Formalizing [Formalizing mathematics in formal system] in Arithmetic] in Lean

𝖱𝟢 では表現定理が成り立つ1．

定理 6 (𝖱𝟢 の表現定理):  Σ1-健全な 𝑇 ⊇ 𝖱𝟢 と，𝑆 を r.e.集合とする． このとき，ℒ︀OR-論理式 
𝖱𝖾𝗉𝑆(𝑥) があって以下を満たす．

𝑛 ∈ 𝑆 ⟺ 𝑇 ⊢ 𝖱𝖾𝗉𝑆(𝑛)

noncomputable def codeOfREPred (p : ℕ → Prop) : ArithmeticSemisentence 1

theorem rePred_weak_representation {p : ℕ → Prop} (hp : REPred p) {x : ℕ} :
    p x ↔ T ⊢ (codeOfREPred p)/[x]

Source: 1

1もちろん Robinson 算術 や 𝖯𝖠 でも成り立つが，𝖱𝟢 は表現定理が成り立つよく知られている算術理論の中では最弱とされる．

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Arithmetic/R0/Representation.lean

表現定理および算術化を用いて，Gödel の第 1 不完全性定理(G1)をまず機械化出来る．

定理 7 (G1):  𝑇  が Cobham の最弱の算術 𝖱𝟢 を含み，Δ1-定義可能で，Σ1-健全なら，𝑇  から
証明も反証も出来ない文が存在する．

証明:  𝐷 ≔ {⌜𝜑⌝ : 𝜑 は 1 変数論理式かつ 𝑇 ⊢ ¬𝜑(⌜𝜑⌝)} を取ると，これは r.e.なので 𝜃(𝑥) 
が存在して 𝑛 ∈ 𝐷 ⟺ 𝑇 ⊢ 𝜃(𝑛)．以下の同値性が成立する．

𝑇 ⊢ 𝜃(⌜𝜃⌝) ⟺ ⌜𝜃⌝ ∈ 𝐷 ⟺ 𝑇 ⊢ ¬𝜃(⌜𝜃⌝)

ゆえに 𝑇  が完全なら無矛盾性に反する． ∎

theorem incomplete (T : ArithmeticTheory) [T.Δ₁] [𝗥₀ ⪯ T] [T.SoundOnHierarchy 𝚺 1] : Incomplete T

Source: 1

ここで Incomplete T は ∃ φ, T ⊬ φ ∧ T ⊬ ∼φ の略記．

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/First.lean

生の証明可能性述語を機械化で直接扱うと面倒なので，抽象化を導入する．

定義 8 (証明可能性):  1 変数述語 𝔅 が 𝑇0 上の 𝑇 -証明可能性 であるとは， 𝐃𝟏 : 𝑇 ⊢ 𝜎 ⟹
𝑇0 ⊢ 𝔅 𝜎 を任意の文 𝜎 で満たすこととする． 追加条件として以下を満たすとき，HBL であ
る1 という．
• 𝐃𝟐 : 𝑇0 ⊢ 𝔅(𝜎 → 𝜏) → (𝔅 𝜎 → 𝔅 𝜏)
• 𝐃𝟑 : 𝑇0 ⊢ 𝔅 𝜎 → 𝔅(𝔅 𝜎)

structure Provability (T₀ : Theory L₀) (T : Theory L) where
  prov : Semisentence L₀ 1
  bew_def {σ : Sentence L} : T ⊢ σ → T₀ ⊢ prov/[⌜σ⌝]

class HBL where
  D2 {σ τ : Sentence L} : T₀ ⊢ 𝔅 (σ 🡒 τ) 🡒 𝔅 σ 🡒 𝔅 τ
  D3 {σ : Sentence L} : T₀ ⊢ 𝔅 σ 🡒 𝔅 (𝔅 σ)

Source: 1

1Hilbert-Bernays-Löb．

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/ProvabilityAbstraction/Basic.lean

定義 9 (対角化可能性):  𝑇  が対角化可能とは，1 変数述語 𝜃 を入力とし文 fixpoint𝜃 を返す関数
があって，それは以下を満たす．

𝑇 ⊢ fixpoint𝜃 ↔ 𝜃(⌜fixpoint𝜃⌝)

class Diagonalization [L.ReferenceableBy L] (T : Theory L) where
  fixedpoint : Semisentence L 1 → Sentence L
  diag (θ) : T ⊢ fixedpoint θ 🡘 θ/[⌜fixedpoint θ⌝]

theorem diagonal (θ : ArithmeticSemisentence 1) :
    T ⊢ fixedpoint θ 🡘 θ/[⌜fixedpoint θ⌝]

Source: 1 2

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/ProvabilityAbstraction/Basic.lean
https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Arithmetic/Bootstrapping/FixedPoint.lean

定義 10:
• 無矛盾性を表す文 ¬ 𝔅 ⊥：「矛盾は証明できない」を Con𝔅 とする．
• ¬ 𝔅(·) の不動点を Gödel 文 G𝔅 とする．

def con (𝔅 : Provability T₀ T) : Sentence L₀ := ∼𝔅 ⊥

def gödel (𝔅 : Provability T₀ T) : Sentence L := fixedpoint T₀ “x. ¬!𝔅.prov x”

lemma gödel_spec : T₀ ⊢ (gödel 𝔅) 🡘 ∼𝔅 (gödel 𝔅)

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/ProvabilityAbstraction/Basic.lean

補題 11 (Abstract G1, G2, Löb):  𝑇  が無矛盾かつ対角化可能で，𝔅 は HBL を満たすとする．
G1 𝑇 ⊬ G𝔅．さらに 𝑇 ⊢ 𝔅 𝜎 ⟹ 𝑇 ⊢ 𝜎 を満たすなら1，𝑇 ⊬ ¬G𝔅．
G2 𝑇 ⊬ Con𝔅
Löb 𝑇 ⊢ 𝔅 𝜎 → 𝜎 なら 𝑇 ⊢ 𝜎
F-Löb 𝑇 ⊢ 𝔅(𝔅 𝜎 → 𝜎) → 𝔅 𝜎

class Kreisel [L.ReferenceableBy L] {T₀ T : Theory L} (𝔅 : Provability T₀ T) where
  KR {σ : Sentence L} : T ⊢ 𝔅 σ → T ⊢ σ

theorem unprovable_gödel : T ⊬ (gödel 𝔅)
theorem unrefutable_gödel [𝔅.Kreisel] : T ⊬ ∼(gödel 𝔅)
theorem con_unprovable [Consistent T] : T ⊬ 𝔅.con
theorem löb_theorem (H : T ⊢ 𝔅 σ 🡒 σ) : T ⊢ σ
theorem formalized_löb_theorem : T₀ ⊢ 𝔅 (𝔅 σ 🡒 σ) 🡒 𝔅 σ

Source: 1

1この条件は Σ1-健全性を純粋に構文論的に仮定したものと見ることが出来る．

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/ProvabilityAbstraction/Basic.lean

算術化を頑張ると HBL を満たす証明可能性を実際に構成することができる．
• 今後この標準的な証明可能性は □𝑇  と書く．

また，対角化可能性も実際満たす． 故に系として，第 2 不完全性定理や Löb の定理を機械化出
来る．

定理 12 (Gödel の第 2 不完全性定理):  𝑇  が 𝖨Σ1 より強く，Δ1-定義可能で，無矛盾なら，𝑇  の
無矛盾性を表す文 ¬□𝑇 ⊥ は証明できない．

theorem consistent_unprovable [Consistent T] : T ⊬ T.consistent.val

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/Second.lean

定理 13 (Löb の定理):  𝑇 ⊢ □𝑇 𝜎 → 𝜎 なら 𝑇 ⊢ 𝜎

theorem löb_theorem : T ⊢ provabilityPred T σ 🡒 σ → T ⊢ σ

Source: 1

𝑇  として Peano 算術 𝖯𝖠 を取ることが出来る．故に 𝖯𝖠 は不完全だし，その無矛盾性は 𝖯𝖠 自
身で証明できない．

instance : 𝗣𝗔 ⪱ 𝗣𝗔 ∪ 𝗣𝗔.Con
instance : 𝗣𝗔 ⪱ 𝗣𝗔 ∪ 𝗣𝗔.Incon

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/Löb.lean

𝔅 としてこのような性質を考えよう

定義 14:  𝐑𝐨𝐬: 𝑇 ⊢ ¬𝜎 ⟹ 𝑇 ⊢ ¬ 𝔅 𝜎

class Rosser [L.ReferenceableBy L₀] {T₀ : Theory L₀} {T : Theory L} (𝔅 : Provability T₀ T) where
  Ros {σ : Sentence L} : T ⊢ ∼σ → T₀ ⊢ ∼𝔅 σ

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/ProvabilityAbstraction/Basic.lean

定理 15 (Abstract GR):  𝑇  が無矛盾かつ対角化可能で 𝔅 が 𝐑𝐨𝐬 を満たすなら， 𝑇 ⊬ G𝔅 か
つ，何も仮定することなく 𝑇 ⊬ ¬G𝔅．

local notation "𝐑" => gödel 𝔅

theorem unrefutable_rosser [𝔅.Rosser] : T ⊬ ∼𝐑
theorem rosser_independent [L.DecidableEq] [𝔅.Rosser] : Independent T 𝐑
theorem rosser_first_incompleteness [L.DecidableEq] (𝔅 : Provability T₀ T) [𝔅.Rosser] :
    Incomplete T

Source: 1

これは Gödel-Rosser の第 1 不完全性定理の抽象的な主張となっている（Σ1-健全性を消去）

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/ProvabilityAbstraction/Basic.lean

実際 witness comparison などの道具を使って 𝐑𝐨𝐬 を満たす証明可能性述語を具体的に構成す
ることが出来る．ゆえに

定理 16 (Gödel-Rosser の第 1 不完全性定理):  𝑇  が 𝖨Σ1 を含み，Δ1-定義可能で，無矛盾な
ら，𝑇  から証明も反証も出来ない文が存在する．

theorem incomplete_GR (T : ArithmeticTheory) [T.Δ₁] [𝗜𝚺₁ ⪯ T] [Consistent T] : Incomplete T

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/RosserProvability.lean

不完全性定理において無矛盾性を表す文は「⊥ が証明されない」以外にもある．
• 例：「任意の文は証明されかつ同時に反証されるということはない」

定理 17 (Jeroslow の第 2 不完全性定理):  𝑇  が無矛盾なら，𝑇 ⊬ ∀𝑥.¬(□𝑇 𝑥 ∧ □𝑇 (¬̇𝑥))

theorem unprovable_formalized_law_of_noncontradiction [𝗜𝚺₁ ⪯ T] [Entailment.Consistent T]
  : T ⊬ ∀¹ ∼(provable T ⋏ T.refutable)

Source: 1

論理式 𝜑 の Gödel 数 ⌜𝜑⌝ を受け取って ¬𝜑 の Gödel 数 ⌜¬𝜑⌝ を返す関数 ¬̇𝑥 を考える必要があ
るが，抽象化ではそれは面倒．
• 反証可能性 𝔚 を考える．つまり，𝑇 ⊢ ¬𝜎 ⟹ 𝑇 ⊢ 𝔚 𝜎 を考えれば良い．

structure Refutability [L.ReferenceableBy L₀] (T₀ : Theory L₀) (T : Theory L) where
  refu : Semisentence L₀ 1
  refu_def {σ : Sentence L} : T ⊢ ∼σ → T₀ ⊢ refu/[⌜σ⌝]

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/Jeroslow.lean
https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/ProvabilityAbstraction/Refutability.lean

各詳細は Saitou and Noguchi (2026) を読んでください（いくつかは次の版で追加予定）．

• Tarski の真理定義不可能性定理．
• Church の一階述語論理の決定不可能性．
•「Gödel 数が 𝑛 以下の証明で証明できる」証明可能性述語について1．
• Ehrenfeucht-Mycielski の加速定理．
• Craig のトリック．
• Friedman–Goldfarb–Harrington 定理（FGH 定理）2．
• 第 1 不完全性定理が成立する理論の Lindenbaum 代数は全て同型．
• IΣ𝑛: 𝑛 ≥ 1 は有限公理化可能．
• 𝖯𝖠 は有限公理化不能（Ryll-Nardzewski の定理）．

1この否定の不動点「Gödel 数が 𝑛 未満の証明では証明できない」は ℕ 上で正しいし，実際に 𝑛 未満の証明で証明できない．故に 𝑛 を途方もなく大
きく取ればそれは正しいが現実的な証明を持たない命題とも言える．

2𝖨Σ1 + 𝖢𝗈𝗇(𝑇 ) 上では任意の Σ1-文は何らかの 𝜎 に対し □𝑇 𝜎 と同値．

𝖤𝖰 𝖱𝟢 𝖰 𝖯𝖠− 𝖨𝖮𝗉𝖾𝗇 𝖨Σ0 𝖨Σ1 𝖡Σ2 𝖨Σ2 𝖯𝖠 𝖯𝖠 + 𝖢𝗈𝗇(𝖯𝖠) 𝖳𝖠

𝖡Σ1

𝖨Σ0 + Ω1 𝖤𝖠
𝖨Σ1 + ¬ 𝖢𝗈𝗇(𝖨Σ1)

𝖨Σ1 + 𝖢𝗈𝗇(𝖨Σ1)

𝖯𝖠 + ¬ 𝖢𝗈𝗇(𝖯𝖠)

𝖯𝖠 + 𝖢𝗈𝗇(𝖯𝖠) + ¬ 𝖢𝗈𝗇(𝖯𝖠 + 𝖢𝗈𝗇(𝖯𝖠))

𝖨Σ+
0

𝖡Π0

𝖨Σ+
1

𝖨Π1

𝖫Σ1

𝖫Π1

𝖡Π1 𝖨Π2

𝖫Σ2

全部 strict だがそこまで出来てない．．．1

1簡単な説明：
• 𝖨Σ1

+ は今回の緩い Σ1 論理式に対しての帰納法原理を追加した体系．
• BΣ1 は Σ1-論理式に対しての採集原理．
• LΣ1 は Σ1-論理式に対しての最小値原理．



4. 証明可能性論理



4. 証明可能性論理
4.1 はじめに 39/92

証明可能性論理の簡単な説明・モチベーション1
• 不完全性定理において中心的な役割を果たす証明可能性述語「𝜑 は 𝑇  で証明できる」を様相

だと思おう．
• 証明可能性 𝔅 をより様相論理的に扱う．

最重要の定理: Solovay の算術的完全性定理．

命題 (Solovay の算術的完全性定理(ラフ)):  HBL な 𝔅 の挙動は様相命題論理 𝐆𝐋 で完全に特
徴づけられる．

theorem arithmetical_completeness_iff [T.SoundOnHierarchy 𝚺 1] :
    𝐆𝐋 ⊢ A ↔ ∀ f : Realization α ℒₒᵣ, T ⊢ f T A

theorem eq_provabilityLogic [T.SoundOnHierarchy 𝚺 1] : 𝐆𝐋 = T.provabilityLogic (α := α)

Source: 1

1証明可能性論理の標準的な文献として Artemov and Beklemishev (2005), Boolos (1994), Smoryński (1985), Verbrugge (2024)．

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/GL/Arithmetic.lean

論理式は ⊥, →, □ を原始的な記号として導入し，あとは略記．

論理式の集合を論理と呼ぶ．

inductive Formula (α : Type*) where
  | atom   : α → Formula α
  | falsum : Formula α
  | imp    : Formula α → Formula α → Formula α
  | box    : Formula α → Formula α
  deriving DecidableEq

abbrev Logic (α : Type*) := Set (Formula α)

Source: 1 2

α として Nat を取れば可算無限個の命題変数，Empty を取れば命題変数がない様相論理式（閉論
理式）を表せる．

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Formula.lean
https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Logic.lean

定義 18:  論理 𝐆𝐋 は古典命題論理に以下の公理と規則を足したもの．
• 規則 (Nec) :⊢ 𝐴 ⟹⊢ □𝐴
• 公理 K : □(𝐴 → 𝐵) → □𝐴 → □𝐵
• 公理 4 : □𝐴 → □□𝐴
• 公理 L : □(□𝐴 → 𝐴) → □𝐴

証明可能性 𝔅 の HBL および F-Löb と以下で対応する．
• 規則 (Nec) が 𝐃𝟏
• 公理 K が 𝐃𝟐
• 公理 4 が 𝐃𝟑
• 公理 L が F-Löb

inductive Logic.normalOf (𝔸 : Set (Formula α)) : Logic α
  | axm {A}        : A ∈ 𝔸 → normalOf 𝔸 A
  | mdp {A B}      : normalOf 𝔸 (A 🡒 B) → normalOf 𝔸 A → normalOf 𝔸 B
  | nec {A}        : normalOf 𝔸 A → normalOf 𝔸 (□A)
  | verum          : normalOf 𝔸 Axioms.Verum
  | implyK {A B}   : normalOf 𝔸 (Axioms.ImplyK A B)
  | implyS {A B C} : normalOf 𝔸 (Axioms.ImplyS A B C)
  | ...
  | dne {A}        : normalOf 𝔸 (Axioms.DNE A)
  | axiomK {A B}   : normalOf 𝔸 (□(A 🡒 B) 🡒 □A 🡒 □B)

abbrev Logic.GL {α : Type*} : Logic α :=
  normalOf ({□A 🡒 □□A | A} ∪ {□(□A 🡒 A) 🡒 □A | A})

Source: 1 2

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Logic.lean
https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/GL/Basic.lean


4. 証明可能性論理
4.3 Kripke 意味論 43/92

実用上はフレームのみを考えることは殆どないので，モデルだけを考えたほうが実装がスッキ
リする．

定義 19 (Kripke モデル):  非空集合 𝑊  とその上の 2 項関係 ≺: 𝑊 × 𝑊 → 2，付値関数 𝑉 :
𝑊 × Prop → 2 の組 ⟨𝑊, ≺, 𝑉 ⟩ を Kripke モデルという．強制関係(Forces)を以下で定める．

• 𝑥 ⊩ 𝑝 ⟺ 𝑥𝑉 𝑝．
• 𝑥 ⊩ □𝐴 ⟺ ∀𝑦, 𝑥 ≺ 𝑦 → 𝑦 ⊩ 𝐴．

Lean では，型 κ 上の構造として定義する．1

1universe を自由にすることが出来ることは完全性定理などでそれなりに問題になるが，無視する．

structure Model (κ : Type*) [Nonempty κ] (α : Type*) where
  Rel' : κ → κ → Prop
  Val' : κ → α → Prop

abbrev World (_ : Model κ α) := κ

abbrev Rel {M : Model κ α} : M.World → M.World → Prop := M.Rel'
scoped infix:60 " ≺ " => Rel

abbrev Val {M : Model κ α} : M.World → α → Prop := M.Val'

def Forces (M : Model κ α) (x : M.World) : Formula α → Prop
  | #a    => M x a
  | ⊥     => False
  | A 🡒 B => Forces M x A → Forces M x B
  | □A    => ∀ y, x ≺ y → Forces M y A

scoped notation:55 x:56 " ⊩[" M "] " A:56 => Forces M x A

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Kripke/Basic.lean

定義 20:
• 𝐆𝐋-モデルとは ≺ が推移的で逆整礎的：𝑥1 ≺ 𝑥2 ≺ … ≺ 𝑥𝑛 が有限の 𝑛 回遷移しか出来ない

とする．
‣ 故に 𝐆𝐋-モデルには点 𝑥 から最大何回遷移できるか：ランク rank(𝑥) が定まる1．

• 有限 𝐆𝐋-モデルとは 𝑊  が有限で ≺ が推移的で非反射的なモデルとする．
• モデルが根付きとは根 𝑟𝑀  があって任意の 𝑥 ∈ 𝑀 \ {𝑟𝑀} に対し 𝑟𝑀 ≺ 𝑥．
‣ 根付きモデルの高さとは rank(𝑟𝑀) とする．

1技術的な面倒のため，Lean での実装では有限モデルのみに対して定めている．

Lean では次のように定義する．

class IsGL (M : Model κ α) extends IsTrans _ M.Rel, IsConverseWellFounded _ M.Rel

class IsFiniteGL (M : Model κ α) extends IsTrans _ M.Rel, Std.Irrefl M.Rel where
  [finite : Finite M.World]

noncomputable def World.rank (x : M.World) : ℕ := cwfHeight (· ≺ ·) x

structure RootedModel (κ : Type*) [Nonempty κ] (α : Type*) extends Model κ α where
  root : toModel.World
  root_rel : ∀ x, x ≠ root → root ≺ x

noncomputable def RootedModel.height (M : RootedModel κ α) [Fintype M.World] [M.IsGL] : ℕ :=
  Model.World.rank (M := M.toModel) M.root

Source: 1 2 3

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Kripke/Basic.lean
https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Kripke/RootedModel.lean
https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Kripke/Rank.lean

余談:

もちろん様相論理一般の議論ではモデルとフレームの区別は非常に重要なのだが，今回は一般
論をしたいわけではないので無視する．

• あくまでも算術を分析するための道具として割り切って実装する．
• 定理証明支援系で何を実装して何を実装しないか．．．
• 道具としての様相論理！

Sambin and Valentini (1982) の 𝐆𝐋 のシークエント計算を機械化する． Kripke 意味論に対す
る完全性の証明はこちらのほうがはるかにスッキリする1．

定義 21:  シークエント Γ ⇒ Δ とは論理式の有限集合の組である． 𝐆𝐋 のシークエント計算 
𝒢︀𝐆𝐋 は次の規則からなる． ただし (WL) と (WR) では Γ ⊆ Γ′，Δ ⊆ Δ′ とする．

(Ax)
𝐴 ⇒ 𝐴

(⊥L)
⊥ ⇒

Γ ⇒ Δ
(WL)

Γ′ ⇒ Δ
Γ ⇒ Δ

(WR)
Γ ⇒ Δ′

Γ ⇒ 𝐴, Δ 𝐵, Γ ⇒ Δ
(→L)

𝐴 → 𝐵, Γ ⇒ Δ
𝐴, Γ ⇒ 𝐵, Δ

(→R)
Γ ⇒ 𝐴 → 𝐵, Δ

□𝐴, Γ, □Γ ⇒ 𝐴
(□𝐆𝐋)

□Γ ⇒ □𝐴

1𝐆𝐋 は標準的な様相論理のカノニカルモデルの構成による証明は不可能．有限反例モデルを直接作るか，適当にモデルを同値で割って証明する．
Hilbert 流だと細かい部分の計算が煩雑になる．

structure Sequent (α : Type*) where
  ant : FormulaFinset α
  suc : FormulaFinset α

infix:50 " ⟹ " => Sequent.mk

inductive Gentzen : Sequent α → Prop
  | axm (A) : Gentzen ({A} ⟹ {A})
  | botL : Gentzen ({⊥} ⟹ ∅)
  | wkL {Γ Γ' Δ} : Gentzen (Γ ⟹ Δ) → (_ : Γ ⊆ Γ' := by grind) → Gentzen (Γ' ⟹ Δ)
  | wkR {Γ Δ Δ'} : Gentzen (Γ ⟹ Δ) → (_ : Δ ⊆ Δ' := by grind) → Gentzen (Γ ⟹ Δ')
  | impL {Γ Δ A B} :
    Gentzen (Γ ⟹ insert A Δ) → Gentzen (insert B Γ ⟹ Δ) → Gentzen (insert (A 🡒 B) Γ ⟹ Δ)
  | impR {Γ Δ A B} : Gentzen (insert A Γ ⟹ insert B Δ) → Gentzen (Γ ⟹ insert (A 🡒 B) Δ)
  | boxGL {Γ A} : Gentzen (insert (□A) (Γ ∪ Γ.box) ⟹ {A}) → Gentzen (Γ.box ⟹ {□A})

notation:45 "⊢ᴳ[𝐆𝐋] " S:50 => Gentzen S

Source: 1 2

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Sequent.lean
https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/GL/Gentzen/Basic.lean

定理 22 (𝒢︀𝐆𝐋 の完全性定理):  𝒢︀𝐆𝐋 ⊢ Γ ⇒ Δ と任意の有限 𝐆𝐋-モデル 𝑀  で 𝑀 ⊨ ⋀ Γ → ⋁ Δ で
あることは同値．

def Model.World.ForcesSequent (M : Model κ α) (x : M.World) (S : Sequent α) : Prop :=
  (∀ C ∈ S.ant, x ⊩ C) → ∃ D ∈ S.suc, x ⊩ D

def Model.ValidateSequent (M : Model κ α) (S : Sequent α) : Prop := ∀ x : M.World, x ⊩ S

lemma GL.Gentzen.iff_valid : ⊢ᴳ[𝐆𝐋] S ↔
    ∀ {κ : Type u} [Nonempty κ] (M : Kripke.Model κ α), [M.IsFiniteGL] → M ⊧ S

Source: 1 2

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/GL/Gentzen/Kripke.lean
https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Kripke/Basic.lean

完全性定理から，カット規則の許容性（意味論的カット除去）がすぐに従う．

定理 23 (𝒢︀𝐆𝐋 のカット除去定理):  カット規則は 𝒢︀𝐆𝐋 で許容される．

Γ1 ⇒ Δ1, 𝐴 𝐴, Γ2 ⇒ Δ2 (Cut)
Γ1, Γ2 ⇒ Δ1, Δ2

theorem GL.Gentzen.cut (h₁ : ⊢ᴳ[𝐆𝐋] Γ₁ ⟹ insert A Δ₁) (h₂ : ⊢ᴳ[𝐆𝐋] insert A Γ₂ ⟹ Δ₂) :
    ⊢ᴳ[𝐆𝐋] Γ₁ ∪ Γ₂ ⟹ Δ₁ ∪ Δ₂

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/GL/Gentzen/Kripke.lean

あとは頑張れば Hilbert 流と対応することがわかる．故に次の同値が言える．

定理 24:  以下同値．𝐴 は論理式．
1. 𝐆𝐋 ⊢ 𝐴
2. 𝒢︀𝐆𝐋 ⊢⇒ 𝐴
3. 任意の有限 𝐆𝐋-モデルで 𝐴 は妥当．
4. 任意の有限 𝐆𝐋-モデルの根で 𝐴 は充足される．
5. 任意の有限 𝐆𝐋-木モデルの根で 𝐴 は充足される．

theorem Logic.GL.provability_TFAE : [
    𝐆𝐋 ⊢ A,
    ⊢ᴳ[𝐆𝐋] ∅ ⟹ {A},
    ∀ {κ : Type u} [Nonempty κ] (M : Model κ α), [M.IsFiniteGL] → M ⊧ A,
    ∀ {κ : Type u} [Nonempty κ] (M : RootedModel κ α), [M.IsFiniteGL] → M.root ⊩ A,
    ∀ {κ : Type u} [Nonempty κ] (M : RootedModel κ α), [M.IsFiniteGL] → [M.IsTree] →
      M.root ⊩ A
  ].TFAE

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/GL/Basic.lean

今回は我々の関心外であるためカット除去は意味論的に行ったが，構文論的なカット除去も可
能である．

シークエントが有限集合ではなくリストや多重集合である場合の構文論的なカット除去は本当
に帰納法が回っていたのか最近まで不明であった．（Goré and Ramanayake (2012) で 3 重帰
納法によって示される．） Brighton (2016) が新しい方法を提案していて，それがうまくいくと
いうことは Goré et al. (2021) が Rocq で検証している．

シークエントを集合で定義しているため，例えば weakening や contraction をどのタイミン
グで行うかは決定的ではなく，（Lean において）現実的に証明探索は出来ない1．

例えば，Maggesi and Perini Brogi (2023) はラベル付きシークエント計算体系を HOL Light
で実装し， GL_provable のような論理式が 𝐆𝐋 で証明できるか判定するタクティクを定義して
いる．

• ただし，このタクティクが必ず停止することは HOL Light の中では証明されていない．数学
的に停止するというメタの保証 Negri (2005) に依存して実装されている．（つまり実装が間
違っていたら永遠に止まらないという可能性はある．）

• 一応我々はラベル付きシークエント計算も実装して停止性も機械化したが．．．

1多分多重集合であったとしても Lean では決定的に証明できないと思う．リストなら出来る気もするが，あまりにも面倒な実装になると思う．

そもそも機械化の上では命題変数 𝑝, 𝑞 として 𝑝 → □𝑞 ではなく，メタの論理式変数 𝐴, 𝐵 として 
𝐴 → □𝐵 は証明可能か？と要請されることがほとんどのため，これらのタクティクはそこまで
役に立たなかった．

実用的には，Kripke 意味論で一旦健全・完全性を使って証明出来る・出来ないを示せば良く，
それでも別にある程度自動化できたりするので十分．

Sambin and Valentini (1982) ではさらにシークエント計算を使った応用として， 𝐆𝐋 の
Craig 補間定理と不動点定理を示している． 𝐆𝐋 の導出木を手作りすれば補間と不動点は構成
的に計算できる(Maehara の方法)12．

定理 25 (𝐆𝐋 の Craig 補間性):  𝐆𝐋 ⊢ 𝐴 → 𝐵 ならば，論理式 𝐶  で 𝐆𝐋 ⊢ 𝐴 → 𝐶  かつ 𝐆𝐋 ⊢
𝐶 → 𝐵 であり，𝐶  の命題変数がすべて 𝐴 と 𝐵 の両方に現れるものが存在する．

theorem Logic.GL.CIP (h : 𝐆𝐋 ⊢ A 🡒 B) :
    ∃ C, 𝐆𝐋 ⊢ A 🡒 C ∧ 𝐆𝐋 ⊢ C 🡒 B ∧ C.atoms ⊆ A.atoms ∩ B.atoms

Source: 1 2

1ただし殆どの場合そんなことはしなくて完全性から作るので意義はない．

2Gignoux (2026) も Lean で補間性定理などを示しているが，意味論的な制約により特殊なケースのみになっている．シークエント計算から一般
に構成出来るというのはそれなりに利点に思える．

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/GL/CIP.lean
https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/GL/Gentzen/Maehara.lean

重要な系として，𝐆𝐋 の補間定理から不動点定理も従う．

定義 26:  命題変数 𝑝 が論理式 𝐴 で modalized であるとは，𝐴 に現れる 𝑝 がすべて □ のス
コープの中にあることをいう．

def ModalizedIn (p : α) : Formula α → Prop
  | #a    => a ≠ p
  | ⊥     => True
  | A 🡒 B => A.ModalizedIn p ∧ B.ModalizedIn p
  | □_    => True

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Formula.lean

定理 27 (𝐆𝐋 の不動点定理 Sambin and Valentini (1982)):  𝑝 が 𝐴 で modalized ならば，𝑝 
を含まず 𝐴 の命題変数のみからなる論理式 𝐷 で

𝐆𝐋 ⊢ 𝐴[𝑝 ≔ 𝐷] ↔ 𝐷

を満たすものが存在する． さらに不動点は証明可能同値を除いて一意である：𝐆𝐋 ⊢ 𝐴[𝑝 ≔
𝐸] ↔ 𝐸 なる任意の論理式 𝐸 について 𝐆𝐋 ⊢ 𝐷 ↔ 𝐸．

theorem Logic.GL.fixpoint_theorem (hpq : p ≠ q) (hA : A.ModalizedIn p) (hq : q ∉ A.atoms) :
    ∃ D, D.atoms ⊆ A.atoms.erase p ∧ 𝐆𝐋 ⊢ A⟦p ↦ D⟧ 🡘 D ∧
      ∀ E, 𝐆𝐋 ⊢ A⟦p ↦ E⟧ 🡘 E → 𝐆𝐋 ⊢ D 🡘 E

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/GL/Fixedpoint.lean

𝐆𝐋 の □ と証明可能性 𝔅 を結びつけよう．

定義 28:  𝑓  を様相論理の命題変数から算術の文への写像とし実現(realization)と呼ぶ． 𝑓  に
よる様相論理式 𝐴 の 𝔅-解釈 𝑓𝔅(𝐴) を以下で定める．
• 𝑓𝔅(𝑝) = 𝑓(𝑝)
• 𝑓𝔅(⊥) = ⊥
• 𝑓𝔅(𝐴 → 𝐵) = 𝑓𝔅(𝐴) → 𝑓𝔅(𝐵)
• 𝑓𝔅(□𝐴) = 𝔅 𝑓𝔅(𝐴)

ただし今後は標準的な証明可能性述語 □𝑇  だけで議論するので 𝑓□𝑇
(𝐴) のみ考える．

structure Realization (α : Type*) (L : Language) where
  val : α → Sentence L

def interpret (f : Realization α L) (𝔅 : Provability T₀ T) : Formula α → Sentence L
  | #a    => f.val a
  | ⊥     => ⊥
  | A 🡒 B => A.interpret f 𝔅 🡒 B.interpret f 𝔅
  | □A    => 𝔅 (A.interpret f 𝔅)

noncomputable abbrev standardInterpret (f : Realization α ℒₒᵣ)
    (T : ArithmeticTheory) [T.Δ₁] : Formula α → Sentence ℒₒᵣ :=
  interpret f T.standardProvability

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Arithmetic/Interpret.lean

定義 29:  算術 𝑇 , 𝑈  とする． 𝑈  上の 𝑇  の証明可能性論理 PL𝑇 (𝑈) を

PL𝑇 (𝑈) ≔ {𝐴 は様相論理式 | 任意の実現 𝑓 に対し 𝑈 ⊢ 𝑓□𝑇
(𝐴)}

で定める．

def ArithmeticTheory.provabilityLogicRelativeTo
    (T U : ArithmeticTheory) [T.Δ₁] : Logic α :=
  { A | ∀ f : Realization α ℒₒᵣ, U ⊢ f T A }

abbrev ArithmeticTheory.provabilityLogic (T : ArithmeticTheory) [T.Δ₁] :
    Logic α :=
  T.provabilityLogicRelativeTo T

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Arithmetic/Interpret.lean

定理 30 (Solovay の算術的完全性定理(Solovay (1976))):  PL𝖯𝖠(𝖯𝖠) = 𝐆𝐋．

theorem arithmetical_completeness_iff [T.SoundOnHierarchy 𝚺 1] :
    𝐆𝐋 ⊢ A ↔ ∀ f : Realization α ℒₒᵣ, T ⊢ f T A

theorem eq_provabilityLogic [T.SoundOnHierarchy 𝚺 1] : 𝐆𝐋 = T.provabilityLogic (α := α)

Source: 1

証明:  健全性 ⟸ は HBL と F-Löb より簡単にわかる． 完全性 ⟹ が難しい．対偶をと
り，反例 𝐆𝐋-木モデルを算術の中に埋め込んで満たさない実現を構成する． ∎

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/GL/Arithmetic.lean

PL𝑇 (𝑈) の 𝑇 , 𝑈  を動かすとどうなるかは Beklemishev (1990) によって完全に分類されている．

定義 31:  以下の論理を定める．𝐆𝐋 + 𝑋 は 𝑋 との union の MP/subst の閉包（非正規拡
大）．
• 𝐆𝐋𝛂(𝑋) ≔ 𝐆𝐋 + {□𝑛+1⊥ → □𝑛⊥ : 𝑛 ∈ 𝑋}．
• 𝐆𝐋𝛃(𝑋) ≔ 𝐆𝐋 + ¬ ⋀

𝑛∉𝑋
(□𝑛+1⊥ → □𝑛⊥)：ただし 𝑋 は補有限．

• 𝐀 ≔ 𝐆𝐋 + {¬□𝑛⊥ : 𝑛 ∈ ℕ}．
• 𝐃 ≔ 𝐆𝐋 + ¬□⊥ + □(□𝐴 ∨ □𝐵) → □𝐴 ∨ □𝐵．
• 𝐒 ≔ 𝐆𝐋 + □𝐴 → 𝐴．

inductive Logic.sumQuasiNormal (L₁ L₂ : Logic α) : Logic α
  | mem₁ {A}    : A ∈ L₁ → sumQuasiNormal L₁ L₂ A
  | mem₂ {A}    : A ∈ L₂ → sumQuasiNormal L₁ L₂ A
  | mdp  {A B}  :
      sumQuasiNormal L₁ L₂ (A 🡒 B) → sumQuasiNormal L₁ L₂ A → sumQuasiNormal L₁ L₂ B
  | subst {A s} : sumQuasiNormal L₁ L₂ A → sumQuasiNormal L₁ L₂ (A⟦s⟧)

infix:50 " +ᴸ " => Logic.sumQuasiNormal

abbrev Logic.S {α : Type*} : Logic α := 𝐆𝐋 +ᴸ { □A 🡒 A | A }
notation "𝐒" => Logic.S

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Logic.lean

定義 32:  論理式 𝐴 のトレース

tr(𝐴) ≔ {𝑛 ∈ ℕ : 𝑟𝑀 ⊮ 𝐴 となる 高さ 𝑛 の有限根付きモデル 𝑀 が存在}

論理 𝐿 のトレース tr(𝐿) ≔ ⋃𝐴∈𝐿 tr(𝐴)．

def Formula.trace (A : Formula α) : Set ℕ :=
  {n | ∃ (κ : Type u) (_ : Nonempty κ) (M : RootedModel κ α)
          (_ : Fintype M.World) (_ : M.IsGL), M.height = n ∧ M.root ⊮ A}

def Logic.trace (L : Logic α) : Set ℕ := ⋃ A ∈ L, A.trace

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Trace.lean

定義 33:  𝑇 ⊢ □𝑛
𝑇 ⊥ となる最小の 𝑛 を 理論 𝑇  の高さ hgt(𝑇 ) という．なければ 𝜔．

noncomputable def Provability.height (𝔅 : Provability T₀ T) : ENat := ENat.find (T ⊢ 𝔅^[·] ⊥)

lemma height_eq_top_iff : 𝔅.height = ⊤ ↔ ∀ n, T ⊬ 𝔅^[n] ⊥

noncomputable abbrev ArithmeticTheory.height (T : ArithmeticTheory) [T.Δ₁] : ℕ∞ :=
  T.standardProvability.height

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/FirstOrder/Incompleteness/ProvabilityAbstraction/Height.lean

定理 34 (Beklemishev (1990)):  𝑇  を 𝖨Σ1 を含む Δ1-定義可能な算術，𝑈  を任意の算術とし，
𝐿 ≔ PL𝑇 (𝑈), 𝑋 ≔ tr(𝐿) とする．
1. 𝑋 が補無限なら 𝐿 = 𝐆𝐋𝛂(𝑋)．
2. 𝑋 が補有限かつ 𝐿 ⊈ 𝐒 なら 𝐿 = 𝐆𝐋𝛃(𝑋)．
3. 𝑋 が補有限かつ 𝐿 ⊆ 𝐒 なら 𝐿 は 𝐆𝐋𝛂(𝑋), 𝐃 ∩ 𝐆𝐋𝛃(𝑋), 𝐒 ∩ 𝐆𝐋𝛃(𝑋) のいずれか．

theorem provabilityLogic_classification :
    letI L := T.provabilityLogicRelativeTo U (α := α);
    L = 𝐆𝐋α L.trace ∨
    ∃ hL : L.traceᶜ.Finite, L = 𝐆𝐋β _ hL ∨ L = 𝐃 ∩ 𝐆𝐋β _ hL ∨ L = 𝐒 ∩ 𝐆𝐋β _ hL

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Classification/General.lean

定理 35 (Beklemishev (1990)):  𝑇  を 𝖨Σ1 を含む Δ1-定義可能な算術とし，真の算術の証明可
能性論理 𝐿 ≔ PL𝑇 (𝖳𝖠) について．
1. 𝑇  が健全なら 𝐿 = 𝐒．
2. 健全ではないが Σ1-健全なら 𝐿 = 𝐃．
3. Σ1-健全でなく hgt(𝑇 ) = 𝜔 なら 𝐿 = 𝐀．
4. hgt(𝑇 ) = 𝑛 < 𝜔 なら 𝐿 = 𝐆𝐋𝛃(ℕ \ {𝑛})．

のいずれか一つのみが成立する．

theorem provabilityLogic_TA_classification : [
    ℕ↓[ℒₒᵣ] ⊧* T ∧ T.provabilityLogicRelativeTo 𝗧𝗔 (α := α) = 𝐒,
    T.SoundOnHierarchy 𝚺 1 ∧ ¬ℕ↓[ℒₒᵣ] ⊧* T ∧ T.provabilityLogicRelativeTo 𝗧𝗔 (α := α) = 𝐃,
    ¬T.SoundOnHierarchy 𝚺 1 ∧ T.height = ⊤ ∧ T.provabilityLogicRelativeTo 𝗧𝗔 (α := α) = 𝐀,
    ∃ n : ℕ, T.height = n ∧ T.provabilityLogicRelativeTo 𝗧𝗔 (α := α) = 𝐆𝐋β {n}ᶜ (by simp)
  ].OAOO

Source: 1

https://github.com/FormalizedFormalLogic/Foundation/blob/f3972f4204fc61e1b736ed843415894c83f35508/Foundation/ProvabilityLogic/Classification/Truth.lean

𝖦𝖫 𝖦𝖫 + ¬□⊥ 𝖠 𝖣 𝖲

𝖦𝖫 + □⊥

PL(𝖯𝖠, 𝖯𝖠) PL(𝖯𝖠, 𝖯𝖠 + Con𝖯𝖠) PL(𝖯𝖠, 𝖯𝖠 + Con𝜔
𝖯𝖠)

PL(𝖨Σ1, 𝖨Σ1 + Con𝜔
𝖨Σ1

)

PL(𝖯𝖠, 𝖯𝖠 + Rfn𝖯𝖠(Σ1))

PL(𝖨Σ1, 𝖨Σ1 + Rfn𝖨Σ1
(Σ1))

PL(𝖯𝖠, 𝖳𝖠)

PL(𝖯𝖠, 𝖯𝖠 + ¬Con𝖯𝖠)

PL(𝖯𝖠 + ¬Con𝖯𝖠, 𝖯𝖠 + ¬Con𝖯𝖠)

原理的には，それぞれの中間には 2𝜔 個の相異な証明可能性論理がある（はず）．

まだまだ山程の課題がある！

• 算術的階層をどう扱う？
• 理論の解釈(interpretation)の実装
• 部分真理定義述語とそれを用いた議論
• もっと弱い算術
• 𝖯𝖠 などの独立命題 (Paris-Harrington の定理など)
• 直観主義(述語論理・算術・様相論理)
• 集合論

などなど．．．

Heyting 算術 𝖧𝖠（直観主義論理上の 𝖯𝖠）の証明可能性論理は 50 年ぐらいの問題1．
• 最近の進展として Mojtahedi (2026) が完全な公理化を発表したが，数年間は査読中2．
• これを機械化して厳密に検証したい．

直観主義様相論理：
• 𝐢𝐆𝐋：直観主義命題論理 + K + (Nec) + □(□𝐴 → 𝐴) → □𝐴．𝖧𝖠 に対して算術的に健全
（Litak (2014), van der Giessen and Iemhoff (2021)）．

• 𝐢𝐒𝐋：𝐢𝐊 + (□𝐴 → 𝐴) → 𝐴（Visser and Litak (2024)）．

Rocq での先行研究（様相論理的な興味）
• 𝐢𝐆𝐋 のカット除去 Goré and Shillito (2022), Shillito (2022)
• 𝐢𝐒𝐋 のカット除去 Shillito et al. (2023)，一様補間定理 Férée et al. (2024)．

1Artemov and Beklemishev (2005), Beklemishev and Visser (2006)

2v1 は 2022 年出版

Heyting 算術をまずどう形式化するか？
• 𝖧𝖠 の証明をどうやって書く？
• 完全性経由で一旦モデルに持ってくるみたいな議論は出来なさそう．

解決案の候補：Lean の中で 𝖧𝖠 のメタ定理証明支援系を作って形式証明を書く？

example : HA proves "∀ x, ∀ y, 0 < x + y + 1" := work_in
  HA_intro;
  ...

公理的集合論の形式化はたまにやられている．

• Lean3 で CH の独立性 (flyspeck プロジェクト: Han and van Doorn (2020))．
‣ ただし Boole 値モデルによるもので，強制法より適用範囲が狭いという問題点がある．

• Isabelle/ZF では，強制法による AC，¬AC や CH の独立性が形式化されている cf: Gunther 
et al. (2020), Gunther et al. (2022), Paulson (2003), 舟根 (2024)．

ただしこれ以上の進展はあまりない（様に見える）．

強制法の一般的な枠組みを機械化して，連続体仮説の独立性などを示したい．
• 方針 1：可算推移的モデル 𝑀  と generic filter 𝐺 から 𝑀[𝐺] を作る標準的な方法 (cf: Kunen 

(2011))．
• 方針 2：証明論的強制法 (cf: Avigad (2004))：理論間の強制解釈と保存性を示す．𝖹𝖥𝖢 の無

矛盾性だけから独立性が従い，証明の変換も有限的．
‣ 一階述語論理の完全性定理の機械化では既に簡略化された強制法を使っている．

その他の問題：半順序などに関する 𝖫𝖾𝖺𝗇 ⊢ 𝜑 の議論をどう 𝖫𝖾𝖺𝗇 ⊢ "𝖹𝖥𝖢 ⊢ 𝜑" へどうやって持っ
ていく？1

1Lean の型理論から考えるとこれを素朴に変換する仕組みが作れるとは思えない．フルスクラッチで書き直す必要もある？

考えたいこと？

• もっと集合論の独立命題．
• 巨大基数の比較：https://neugierde.github.io/cantors-attic/．
• 強制法の様相論理：強制法の原理は様相論理 𝐒𝟒.𝟐 の拡張体系でおよそ記述出来る．

https://neugierde.github.io/cantors-attic/

現状，Formalized Formal Logic の今回の形式化は Mathlib へ移植する気はほとんどない．

OSS 開発としての問題
• Mahtlib のレビューは遅く，また論理学の適切なレビュアーが今いない．
• Mathlib の肥大化：ライブラリが 8GB もあるのどうなの？

論理学特有の問題
• 論理学はどのような言語や定式化を選択するかなどの定義に強く依存する．
‣ そのようなものを汎用的なライブラリとして提供する（数学的・技術的）方法がない，ある

いは意義が薄い．

今後はあくまでも FFL のフレームワークで論理学をやる方面になるだろう（デファクト・スタ
ンダードになればいいな）

Formal Conjectures1 という Lean による未解決問題のカタログがある．これの FFL 版を作
ろう．

例えば算術には以下の問題がある
1. IΔ0 は有限公理化可能か？
2. IΔ0 で素数が無限であることは証明可能か？

このような問題をカタログ化しておけば誰か（何か）が取り組んでくれるかも？2

1https://github.com/google-deepmind/formal-conjectures

2それが果たしてどうなんだ？という問題はありますが．

https://github.com/google-deepmind/formal-conjectures

注意 (再掲):
1. 依拠している数学的基礎に問題や矛盾が無いか？（論理学者として）
2. ソフトウェアとしての実装にバグが存在しないか？（ソフトウェア開発者として）
3. 形式化したそのコードは命題を正しく形式化しているか？（エンドユーザとして）

「定理証明支援系を開発する」という目的でなかったとしても，形式証明を書くという事業はソ
フトウェア開発の一種である．

OpenAI, Navier-Stokes 方程式に関する形式化のコミットの例．

（Anthoropic の FLT の定理の形式化はまず GitHub のプレビューが追いついていなかった）

OpenAI や Anthoropic などが提出した NS 方程式や FLT の形式証明は問題があると個人的に
は思う．

妥当性の問題 本当に正しく主張を形式化出来てる？
• 100k 行の証明をどうやって一気に人間がレビューしろと？
• この膨大なコードの中で定理証明支援系自体の実装のバグを踏んでいないとなぜ言える？

重要度の問題 アドホックな補題や定義・議論が多すぎるのでは？
• 1 回しか使わない補題など，本当に切り出すべきか？

再利用性の問題 車輪の再発明では？
• 難問にはそれなりにたくさんの道具が必要，それらを毎回毎回用意し直すのは無駄．
• 適切にライブラリとして切り出せばコミュニティとしても嬉しいはず．しかし一切その

ようなことが行われている気配はない1．

1ただし現状では Mathlib は AI/LLM による PR は受け付けていない．それでは困るので，最近では Tau Ceti Project https://github.com/
TauCetiProject/TauCeti という AI の使用やレビューを受け付けている実験場も用意されている．

https://github.com/TauCetiProject/TauCeti
https://github.com/TauCetiProject/TauCeti

定理証明支援系・形式証明を
• 単なる自然言語による証明のより精緻な保証の執筆作業としてではなく，
• 数学をより高速かつ正確に生産するソフトウェア開発の視点を投げかける：証明工学

一般のソフトウェア開発にはあまりない面白い課題が多く眠っていると思う．

deformalization 逆方向，つまり，既にある形式証明から逆に自然言語で書かれた読みやすい
証明へと翻訳する．あるいは，そもそもこの形式証明が書かれた意図などを補題から抽出
することで，新たな数学的理論の創出に繋がる可能性がある．

visualization 定理や補題同士の繋がりの可視化や，そもそも数学的構造自体の可視化を自動で
行えたら，それは形式証明を人力で読み書きする人間にも役立つだろう．

collaborating 数学というかなり専門性や属人性の高い知識が要求される開発者をどのように
マネジメントしたりプロジェクトを進めるべきか？

architecture もっと効率的・正確に形式証明を生成するような自動証明アーキテクチャはどの
ようなものだろうか？

もちろん古典的なソフトウェア工学的な知見だって大いに役立つだろう1．

1ソフトウェア工学という学問が現代ないし現場のソフトウェア開発に果たしてどれだけ寄与したのか，あるいは単に机上の空論ではなかったのか？
という批判はありますが．

Solovay の算術的完全性定理までは LLM は積極的に利用していなかった（2025 年ごろ）．

積極的に利用し始めたのは証明可能性論理の分類定理を機械化するにあたって（2026 年 7 月）． 
技術的に面倒な部分があって停滞していたが，大きく加速した．

Kripke モデルの幾何的な議論（絵で書いたら自明じゃん）をどう Lean でコードを書くべき
か？が難しく（あるいはただ面倒な議論で）詰まっていた．

例:  モデル 𝑀  の適当な点 𝑎 の前に 𝑘 ≤ 𝜔 個の点を追加する． このとき，新しいモデル 𝑀 ′ の
高さ hgt(𝑀) + 𝑘 は有限個の追加ならせいぜい hgt(𝑀) + 𝑘．

𝑎
𝑀

hgt(𝑀) = ℎ

⇓

⋯
𝑎

𝑀 ′

𝑘

hgt(𝑀 ′) = ℎ + 𝑘

機械化するとなると結構面倒な議論になる．

AI/LLM は Mathlib の全てのコードを学習しており(おそらく)，自分が知らなかったようなタ
クティクや方法の形式証明を書いてくれる．

個人的には，来年(2027 年)にはもうそれほど人力で形式証明を書かなくても良い時代が来るの
かなとは思う．
• 便利なキーボードとしての AI/LLM．
• もちろん全体的な議論や実装の筋の良さ，みたいなものは人が評価するべきだとは思う．

最近では AlphaCentauri というリポジトリで AI/LLM 生成した算術周りの証明を置いていま
す．
• https://github.com/FormalizedFormalLogic/AlphaCentauri
• 人間半分，AI 半分で自動証明．
‣ 人間の担当箇所：証明の方針や議論全体の設計や切り分け．PR のレビュー．
‣ AI の担当箇所：人間の方針をもとに実際のコード量の見積もりや形式証明の執筆．

レビュー出来ていないが外部の協力者から 𝖯𝖠 の独立命題の形式化も送られてきている．
• https://github.com/FormalizedFormalLogic/goodstein-independence
• Goodstein 列の停止性，Hydra ゲーム，Paris-Harrignton の定理など．

https://github.com/FormalizedFormalLogic/AlphaCentauri
https://github.com/FormalizedFormalLogic/goodstein-independence

この発表では定理証明支援系 Lean で Gödel の不完全性定理と証明可能性論理に関する事実を
形式化したということを発表しました．

定理証明支援系は楽しいので皆さんも触ってみてください．
• 将来は人間のプレプリントを全部形式化して厳密に追試することが出来る，みたいな世界が

できたらそれは素晴らしいことだと思います．
• ロジックの形式化をしている人や資産が本当に足りてないので，知的資源およびソフトウェ

ア的な資源で協力してくれる人がいたら本当に助かります．



5. まとめと今後の展望と余談
5.8 参考文献 89/92

説明出来なかった機械化の技術的な詳細は 齋藤 (2024) にも記載がある（それほど実装の本筋
は変わっていない）．

@Alwe_Logic, 2026. https://x.com/Alwe_Logic/status/2099488711427973188 [WWW Document].. URL https://x.com/Alwe_
Logic/status/2099488711427973188

Artemov, S.N., Beklemishev, L.D., 2005. Provability Logic, in: Gabbay, D.M., Guenthner, F. (Eds.), Handbook of Philosophical Logic, 
2nd Edition. Springer Netherlands, Dordrecht, pp. 189–360.. https://doi.org/10.1007/1-4020-3521-7_3

Avigad, J., 2004. Forcing in Proof Theory. The Bulletin of Symbolic Logic 10, 305–333.. https://doi.org/10.2178/bsl/
1102022660

Avigad, J., 2001. Algebraic Proofs of Cut Elimination. The Journal of Logic and Algebraic Programming 49, 15–30.. https://doi.
org/10.1016/S1567-8326(01)00009-1

Beklemishev, L.D., 1990. On the Classification of Propositional Provability Logics. Mathematics of the USSR-Izvestiya 35, 247–275.. 
https://doi.org/10.1070/IM1990v035n02ABEH000701

Beklemishev, L.D., Visser, A., 2006. Problems in the Logic of Provability, in: Gabbay, D.M., Goncharov, S.S., Zakharyaschev, M. 
(Eds.), Mathematical Problems from Applied Logic I. Springer, New York, pp. 77–136.. https://doi.org/10.1007/0-387-
31072-X_2

Boolos, G., 1994. The Logic of Provability. Cambridge University Press, Cambridge.. https://doi.org/10.1017/CBO
9780511625183

Brighton, J., 2016. Cut Elimination for GLS Using the Terminability of Its Regress Process. Journal of Philosophical Logic 45, 147–
153.. https://doi.org/10.1007/s10992-015-9368-4

https://x.com/Alwe_Logic/status/2099488711427973188
https://x.com/Alwe_Logic/status/2099488711427973188
https://doi.org/10.1007/1-4020-3521-7_3
https://doi.org/10.2178/bsl/1102022660
https://doi.org/10.2178/bsl/1102022660
https://doi.org/10.1016/S1567-8326(01)00009-1
https://doi.org/10.1016/S1567-8326(01)00009-1
https://doi.org/10.1070/IM1990v035n02ABEH000701
https://doi.org/10.1007/0-387-31072-X_2
https://doi.org/10.1007/0-387-31072-X_2
https://doi.org/10.1017/CBO9780511625183
https://doi.org/10.1017/CBO9780511625183
https://doi.org/10.1007/s10992-015-9368-4


5. まとめと今後の展望と余談
5.8 参考文献 90/92

Carneiro, M., 2019. The Type Theory of Lean (Master's thesis).

Férée, H., van der Giessen, I., van Gool, S., Shillito, I., 2024. Mechanised Uniform Interpolation for Modal Logics K, GL, and iSL, in: 
Benzmüller, C., Heule, M.J.H., Schmidt, R.A. (Eds.), Automated Reasoning. Springer Nature Switzerland, Cham, pp. 43–60.. 
https://doi.org/10.1007/978-3-031-63501-4_3

Gignoux, M., 2026. Proofs as Coalgebras (Master's thesis).

Goré, R., Ramanayake, R., 2012. Valentini's Cut-Elimination for Provability Logic Resolved. The Review of Symbolic Logic 5, 212–
238.. https://doi.org/10.1017/S1755020311000323

Goré, R., Ramanayake, R., Shillito, I., 2021. Cut-Elimination for Provability Logic by Terminating Proof-Search: Formalised and 
Deconstructed Using Coq, in: Das, A., Negri, S. (Eds.), Automated Reasoning with Analytic Tableaux and Related Methods. 
Springer International Publishing, Cham, pp. 299–313.. https://doi.org/10.1007/978-3-030-86059-2_18

Goré, R., Shillito, I., 2022. Direct Elimination of Additive-Cuts in GL4ip: Verified and Extracted, in: Fernández-Duque, D., Palmigiano, 
A., Pinchinat, S. (Eds.), Advances in Modal Logic. College Publications, London, pp. 429–450.

Gunther, E., Pagano, M., Sánchez Terraf, P., 2020. Formalization of Forcing in Isabelle/ZF, in: Peltier, N., Sofronie-Stokkermans, V. 
(Eds.), Automated Reasoning. Springer International Publishing, Cham, pp. 221–235.. https://doi.org/10.1007/978-3-030-
51054-1_13

Gunther, E., Pagano, M., Sánchez Terraf, P., Steinberg, M., 2022. The Independence of the Continuum Hypothesis in Isabelle/ZF 
[WWW Document].. URL https://isa-afp.org/entries/Independence_CH.html

Han, J.M., van Doorn, F., 2020. A Formal Proof of the Independence of the Continuum Hypothesis. Association for Computing 
Machinery, New Orleans, LA, USA, pp. 353–366.. https://doi.org/10.1145/3372885.3373826

Kunen, K., 2011. Set Theory. College Publications, London.

https://doi.org/10.1007/978-3-031-63501-4_3
https://doi.org/10.1017/S1755020311000323
https://doi.org/10.1007/978-3-030-86059-2_18
https://doi.org/10.1007/978-3-030-51054-1_13
https://doi.org/10.1007/978-3-030-51054-1_13
https://isa-afp.org/entries/Independence_CH.html
https://doi.org/10.1145/3372885.3373826


5. まとめと今後の展望と余談
5.8 参考文献 91/92

Litak, T., 2014. Constructive Modalities with Provability Smack, in: Bezhanishvili, G. (Ed.), Leo Esakia on Duality in Modal and 
Intuitionistic Logics. Springer Netherlands, Dordrecht, pp. 187–216.. https://doi.org/10.1007/978-94-017-8860-1_8

Maggesi, M., Perini Brogi, C., 2023. Mechanising Gödel–Löb Provability Logic in HOL Light. Journal of Automated Reasoning 67.. 
https://doi.org/10.1007/s10817-023-09677-z

Mojtahedi, M., 2026. On the Provability Logic of HA.. https://doi.org/10.48550/arXiv.2206.00445

Moura, L. de, 2026. Postmortem for Kernel Soundness Bug #14576.

Negri, S., 2005. Proof Analysis in Modal Logic. Journal of Philosophical Logic 34, 507–544.. https://doi.org/10.1007/s10992-
005-2267-3

O'Connor, R., 2005. Essential Incompleteness of Arithmetic Verified by Coq, in: Hurd, J., Melham, T. (Eds.), Theorem Proving in 
Higher Order Logics. Springer Berlin Heidelberg, Berlin, Heidelberg, pp. 245–260.. https://doi.org/10.1007/11541868_16

Paulson, L.C., 2014. A Machine-Assisted Proof of Gödel's Incompleteness Theorems for the Theory of Hereditarily Finite Sets. The 
Review of Symbolic Logic 7, 484–498.. https://doi.org/10.1017/S1755020314000112

Paulson, L.C., 2003. The Relative Consistency of the Axiom of Choice Mechanized Using Isabelle/ZF. LMS Journal of Computation 
and Mathematics 6, 198–248.. https://doi.org/10.1112/S1461157000000449

Saitou, S., Noguchi, M., 2026. Mechanizing Gödel's Incompleteness Theorems and Provability Logic.

Sambin, G., Valentini, S., 1982. The Modal Logic of Provability. The Sequential Approach. Journal of Philosophical Logic 11, 311–
342.. https://doi.org/10.1007/BF00293433

Shankar, N., 1986. Proof-Checking Metamathematics (Theorem-Proving) (Ph.D. thesis).

Shillito, I., 2022. New Foundations for the Proof Theory of Bi-Intuitionistic and Provability Logics Mechanized in Coq (Ph.D. thesis).

https://doi.org/10.1007/978-94-017-8860-1_8
https://doi.org/10.1007/s10817-023-09677-z
https://doi.org/10.48550/arXiv.2206.00445
https://doi.org/10.1007/s10992-005-2267-3
https://doi.org/10.1007/s10992-005-2267-3
https://doi.org/10.1007/11541868_16
https://doi.org/10.1017/S1755020314000112
https://doi.org/10.1112/S1461157000000449
https://doi.org/10.1007/BF00293433


5. まとめと今後の展望と余談
5.8 参考文献 92/92

Shillito, I., van der Giessen, I., Goré, R., Iemhoff, R., 2023. A New Calculus for Intuitionistic Strong Löb Logic: Strong Termination and 
Cut-Elimination, Formalised, in: Ramanayake, R., Urban, J. (Eds.), Automated Reasoning with Analytic Tableaux and Related 
Methods. Springer Nature Switzerland, Cham, pp. 73–93.. https://doi.org/10.1007/978-3-031-43513-3_5

Smoryński, C., 1985. Self-Reference and Modal Logic, Universitext. Springer New York, New York, NY.. https://doi.org/10.1007/
978-1-4613-8601-8

Solovay, R.M., 1976. Provability Interpretations of Modal Logic. Israel Journal of Mathematics 25, 287–304.. https://doi.org/10.
1007/BF02757006

van der Giessen, I., Iemhoff, R., 2021. Sequent Calculi for Intuitionistic Gödel–Löb Logic. Notre Dame Journal of Formal Logic 62, 
221–246.. https://doi.org/10.1215/00294527-2021-0011

Verbrugge, R., 2024. Provability Logic [WWW Document]. The Stanford Encyclopedia of Philosophy (Summer 2024 Edition).. URL 
https://plato.stanford.edu/archives/sum2024/entries/logic-provability/ (accessed 7.27.26).

Visser, A., Litak, T., 2024. Lewis and Brouwer Meet Strong Löb.. https://doi.org/10.48550/arXiv.2404.11969

舟根大喜, 2024. ZF+¬AC の相対的無矛盾性証明の Isabelle/ZF による形式化 [WWW Document].. URL https://www.math.nagoya-u.
ac.jp/~garrigue/tpp2024/slides/funane.pdf

